// File: contracts/RollStakingRewards.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity =0.7.6;

import "./openzeppelin/utils/ReentrancyGuard.sol";
import "./openzeppelin/math/Math.sol";
import "./openzeppelin/token/ERC20/IERC20.sol";
import "./openzeppelin/math/SafeMath.sol";
import "./openzeppelin/utils/Address.sol";
import "./openzeppelin/token/ERC20/SafeERC20.sol";

import "./RollRewardsDistributionRecipient.sol";
import "./TokenWrapper.sol";

contract RollStakingRewards is
	RollRewardsDistributionRecipient,
	TokenWrapper,
	ReentrancyGuard
{
	using SafeMath for uint256;
	using SafeERC20 for IERC20;
	/* ========== STATE VARIABLES ========== */

	struct TokenRewardData {
		uint256 rewardRate;
		uint256 rewardPerTokenStored;
	}

	address[] public rewardTokensAddresses;
	mapping(address => TokenRewardData) public rewardTokens;
	mapping(address => mapping(address => uint256))
		public userRewardPerTokenPaid;
	mapping(address => mapping(address => uint256)) public rewards;
	mapping(address => uint256) public freeTokens;

	/* ========== GLOBAL STATE VARIABLES ==========  */

	uint256 public periodStart;
	uint256 public periodFinish;
	uint256 public rewardsDuration;

	uint256 public lastUpdateTime;
	uint256 public immutable tokenDecimals;

	uint256 public lastUnstake;

	/* ========== CONSTRUCTOR ========== */

	constructor(
		address _owner,
		address _rewardsDistribution,
		address[] memory _rewardTokens,
		address _stakingToken,
		address _registry
	) RollOwned(_owner, _registry) {
		for (uint256 i = 0; i < _rewardTokens.length; i++) {
			rewardTokensAddresses.push(_rewardTokens[i]);
			rewardTokens[_rewardTokens[i]] = TokenRewardData(0, 0);
		}
		token = IERC20(_stakingToken);
		tokenDecimals = token.decimals();
		rewardsDistribution = _rewardsDistribution;
	}

	/* ========== VIEWS ========== */

	function lastTimeRewardApplicable() public view returns (uint256) {
		uint256 n = Math.min(block.timestamp, periodFinish);
		return Math.max(n, periodStart);
	}

	function isValidRewardToken(address _token) public view returns (bool) {
		for (uint256 i = 0; i < rewardTokensAddresses.length; i++)
			if (_token == address(rewardTokensAddresses[i])) return true;
		return false;
	}

	function rewardPerToken(address _token) public view returns (uint256) {
		TokenRewardData storage data = rewardTokens[_token];
		if (_totalSupply == 0 || block.timestamp < periodStart) {
			return data.rewardPerTokenStored;
		}

		return
			data.rewardPerTokenStored.add(
				lastTimeRewardApplicable()
					.sub(lastUpdateTime)
					.mul(data.rewardRate)
					.mul(10**tokenDecimals)
					.div(totalSupply())
			);
	}

	function earned(address _account, address _token)
		public
		view
		returns (uint256)
	{
		return
			balanceOf(_account)
				.mul(
					rewardPerToken(_token).sub(
						userRewardPerTokenPaid[_token][_account]
					)
				)
				.div(10**tokenDecimals)
				.add(rewards[_token][_account]);
	}

	function getRewardForDuration(address _token)
		public
		view
		returns (uint256)
	{
		TokenRewardData storage data = rewardTokens[_token];
		return data.rewardRate.mul(rewardsDuration);
	}

	function getFreeTokenAmount(address token) public view returns (uint256) {
		if (block.timestamp <= periodStart) return 0;
		if (lastUnstake > Math.min(periodFinish, block.timestamp))
			return freeTokens[token];

		uint256 leftover = 0;
		if (super.totalSupply() == 0) {
			uint256 remaining = Math.min(periodFinish, block.timestamp).sub(
				lastUnstake
			);
			TokenRewardData storage data = rewardTokens[token];
			leftover = remaining.mul(data.rewardRate);
		}
		return freeTokens[token].add(leftover);
	}

	/* ========== MUTATIVE FUNCTIONS ========== */
	function moveFreeTokenToCreator() internal {
		uint256 from = Math.max(lastUnstake, periodStart);
		uint256 to = Math.min(block.timestamp, periodFinish);
		if (from >= to) return;
		if (super.totalSupply() == 0) {
			uint256 remaining = to.sub(from);
			for (uint256 i = 0; i < rewardTokensAddresses.length; i++) {
				TokenRewardData storage data = rewardTokens[
					rewardTokensAddresses[i]
				];
				uint256 leftover = remaining.mul(data.rewardRate);
				freeTokens[rewardTokensAddresses[i]] = freeTokens[
					rewardTokensAddresses[i]
				].add(leftover);
			}
		}
		lastUnstake = Math.max(block.timestamp, lastUnstake);
	}

	function stake(uint256 amount)
		public
		override
		nonReentrant
		notPaused
		updateReward(msg.sender)
	{
		require(amount > 0, "Cannot stake 0");
		moveFreeTokenToCreator();
		super.stake(amount);
		emit Staked(msg.sender, amount);
	}

	function withdraw(uint256 amount)
		public
		override
		nonReentrant
		updateReward(msg.sender)
	{
		require(amount > 0, "Cannot withdraw 0");
		moveFreeTokenToCreator();
		super.withdraw(amount);
		emit Withdrawn(msg.sender, amount);
	}

	function getReward() public nonReentrant updateReward(msg.sender) {
		for (uint256 i = 0; i < rewardTokensAddresses.length; i++) {
			uint256 reward = rewards[address(rewardTokensAddresses[i])][
				msg.sender
			];
			if (reward > 0) {
				rewards[address(rewardTokensAddresses[i])][msg.sender] = 0;
				IERC20(rewardTokensAddresses[i]).safeTransfer(
					msg.sender,
					reward
				);
				emit RewardPaid(
					msg.sender,
					address(rewardTokensAddresses[i]),
					reward
				);
			}
		}
	}

	function exit() external {
		withdraw(_balances[msg.sender]);
		getReward();
	}

	/* ========== RESTRICTED FUNCTIONS ========== */

	/* issues with adding multiple tokens
	 * Either we leave them and there will be wasted gas for expired rewards or
	 * we remove a token from the array and prevent anyone from claiming if they forgot to do so
	 * An announcement may be given a week prior from removing a secondary reward token.
	 * this would keep calls as efficient as possible.
	 */
	function notifyRewardAmount(
		uint256[] calldata _rewards,
		address[] calldata _tokens
	)
		external
		override
		onlyRewardsDistribution
		updateReward(address(0))
		lazyUpdateReward(address(0))
	{
		require(
			block.timestamp > periodFinish,
			"RollStakingRewards: Previous rewards period must be complete before changing the duration for the new period"
		);
		require(
			periodStart > 0 && block.timestamp < periodStart,
			"RollStakingRewards: Period not set"
		);
		require(
			_rewards.length == _tokens.length &&
				rewardTokensAddresses.length == _tokens.length,
			"RollStakingRewards: Amount of rewards not matching reward token count."
		);
		moveFreeTokenToCreator();
		for (uint256 i = 0; i < _tokens.length; i++) {
			uint256 balance = 0;
			TokenRewardData storage data = rewardTokens[_tokens[i]];

			data.rewardRate = _rewards[i].div(rewardsDuration);

			//==============================
			// transfer the tokens
			IERC20 _token = IERC20(_tokens[i]);
			bool ok = _token.transferFrom(
				msg.sender,
				address(this),
				_rewards[i]
			);
			require(ok, "LM: no enough tokens");
			uint256 leftover = _rewards[i].sub(
				data.rewardRate.mul(rewardsDuration)
			);
			freeTokens[_tokens[i]] = freeTokens[_tokens[i]].add(leftover);

			balance = _rewards[i];
			//==============================
			require(
				data.rewardRate <= balance.div(rewardsDuration),
				"RollStakingRewards: Provided reward too high"
			);
			emit RewardAdded(_rewards[i], _tokens[i]);
		}
		lastUpdateTime = periodStart;
		periodFinish = periodStart.add(rewardsDuration);
	}

	function setRewardsDuration(uint256 _periodStart, uint256 _rewardsDuration)
		external
		onlyOwner
		updateReward(address(0))
		lazyUpdateReward(address(0))
	{
		require(
			block.timestamp > periodFinish,
			"RollStakingRewards: Previous rewards period must be complete before changing the duration for the new period"
		);
		require(
			block.timestamp < _periodStart || _periodStart == 0,
			"RollStakingRewards: Start must be a future date"
		);

		// if is not the first campaing let's reset reaward balances
		if (periodStart > 0) {
			claimFreeTokensImpl();
		}

		if (_periodStart == 0) {
			periodStart = block.timestamp;
		} else {
			periodStart = _periodStart;
		}

		periodFinish = 0;
		lastUpdateTime = periodStart;
		lastUnstake = periodStart;
		rewardsDuration = _rewardsDuration;
		emit RewardsUpdated(periodStart, rewardsDuration);
	}

	function claimFreeTokensImpl() internal {
		require(
			block.timestamp > periodFinish,
			"RollStakingRewards: Previous rewards period must be complete before claim"
		);

		// check if there's some pending calculation
		moveFreeTokenToCreator();

		// do the transfer
		for (uint256 i = 0; i < rewardTokensAddresses.length; i++) {
			address token = rewardTokensAddresses[i];
			uint256 freeTokenAmount = freeTokens[token];
			freeTokens[token] = 0;
			if (freeTokenAmount > 0) {
				IERC20 _token = IERC20(token);
				bool ok = _token.transfer(msg.sender, freeTokenAmount);
				require(ok, "LM: no enough tokens");
			}
		}

		emit RewardsUpdated(periodStart, rewardsDuration);
	}

	function claimFreeTokens() external onlyOwner {
		claimFreeTokensImpl();
	}

	/* ========== MODIFIERS ========== */
	modifier lazyUpdateReward(address _account) {
		_;
		updateRewardImpl(_account);
	}

	modifier updateReward(address _account) {
		updateRewardImpl(_account);
		_;
	}

	function updateRewardImpl(address _account) internal {
		for (uint256 i = 0; i < rewardTokensAddresses.length; i++) {
			TokenRewardData storage data = rewardTokens[
				rewardTokensAddresses[i]
			];
			data.rewardPerTokenStored = rewardPerToken(
				rewardTokensAddresses[i]
			);
		}
		lastUpdateTime = lastTimeRewardApplicable();
		for (uint256 i = 0; i < rewardTokensAddresses.length; i++) {
			TokenRewardData storage data = rewardTokens[
				rewardTokensAddresses[i]
			];
			if (_account != address(0)) {
				rewards[rewardTokensAddresses[i]][_account] = earned(
					_account,
					rewardTokensAddresses[i]
				);
				userRewardPerTokenPaid[rewardTokensAddresses[i]][
					_account
				] = data.rewardPerTokenStored;
			}
		}
	}

	event RewardAdded(uint256 reward, address indexed token);
	event Staked(address indexed user, uint256 amount);
	event Withdrawn(address indexed user, uint256 amount);
	event RewardPaid(
		address indexed user,
		address indexed token,
		uint256 reward
	);
	event RewardsUpdated(uint256 newStart, uint256 newDuration);
	event Recovered(address token, uint256 amount);
}

// File: contracts/openzeppelin/utils/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
	// Booleans are more expensive than uint256 or any type that takes up a full
	// word because each write operation emits an extra SLOAD to first read the
	// slot's contents, replace the bits taken up by the boolean, and then write
	// back. This is the compiler's defense against contract upgrades and
	// pointer aliasing, and it cannot be disabled.

	// The values being non-zero value makes deployment a bit more expensive,
	// but in exchange the refund on every call to nonReentrant will be lower in
	// amount. Since refunds are capped to a percentage of the total
	// transaction's gas, it is best to keep them low in cases like this one, to
	// increase the likelihood of the full refund coming into effect.
	uint256 private constant _NOT_ENTERED = 1;
	uint256 private constant _ENTERED = 2;

	uint256 private _status;

	constructor() {
		_status = _NOT_ENTERED;
	}

	/**
	 * @dev Prevents a contract from calling itself, directly or indirectly.
	 * Calling a `nonReentrant` function from another `nonReentrant`
	 * function is not supported. It is possible to prevent this from happening
	 * by making the `nonReentrant` function external, and make it call a
	 * `private` function that does the actual work.
	 */
	modifier nonReentrant() {
		// On the first call to nonReentrant, _notEntered will be true
		require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

		// Any calls to nonReentrant after this point will fail
		_status = _ENTERED;

		_;

		// By storing the original value once again, a refund is triggered (see
		// https://eips.ethereum.org/EIPS/eip-2200)
		_status = _NOT_ENTERED;
	}
}

// File: contracts/openzeppelin/math/Math.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a >= b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow, so we distribute
        return (a / 2) + (b / 2) + ((a % 2 + b % 2) / 2);
    }
}

// File: contracts/openzeppelin/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.7.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
	/**
	 * @dev Returns the amount of tokens in existence.
	 */
	function totalSupply() external view returns (uint256);

	/**
	 * @dev Returns the amount of tokens owned by `account`.
	 */
	function balanceOf(address account) external view returns (uint256);

	/**
	 * @dev Moves `amount` tokens from the caller's account to `recipient`.
	 *
	 * Returns a boolean value indicating whether the operation succeeded.
	 *
	 * Emits a {Transfer} event.
	 */
	function transfer(address recipient, uint256 amount)
		external
		returns (bool);

	/**
	 * @dev Returns the remaining number of tokens that `spender` will be
	 * allowed to spend on behalf of `owner` through {transferFrom}. This is
	 * zero by default.
	 *
	 * This value changes when {approve} or {transferFrom} are called.
	 */
	function allowance(address owner, address spender)
		external
		view
		returns (uint256);

	/**
	 * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
	 *
	 * Returns a boolean value indicating whether the operation succeeded.
	 *
	 * IMPORTANT: Beware that changing an allowance with this method brings the risk
	 * that someone may use both the old and the new allowance by unfortunate
	 * transaction ordering. One possible solution to mitigate this race
	 * condition is to first reduce the spender's allowance to 0 and set the
	 * desired value afterwards:
	 * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
	 *
	 * Emits an {Approval} event.
	 */
	function approve(address spender, uint256 amount) external returns (bool);

	/**
	 * @dev Moves `amount` tokens from `sender` to `recipient` using the
	 * allowance mechanism. `amount` is then deducted from the caller's
	 * allowance.
	 *
	 * Returns a boolean value indicating whether the operation succeeded.
	 *
	 * Emits a {Transfer} event.
	 */
	function transferFrom(
		address sender,
		address recipient,
		uint256 amount
	) external returns (bool);

	function decimals() external view returns (uint8);

	/**
	 * @dev Emitted when `value` tokens are moved from one account (`from`) to
	 * another (`to`).
	 *
	 * Note that `value` may be zero.
	 */
	event Transfer(address indexed from, address indexed to, uint256 value);

	/**
	 * @dev Emitted when the allowance of a `spender` for an `owner` is set by
	 * a call to {approve}. `value` is the new allowance.
	 */
	event Approval(
		address indexed owner,
		address indexed spender,
		uint256 value
	);
}

// File: contracts/openzeppelin/math/SafeMath.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.7.0;

/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
	/**
	 * @dev Returns the addition of two unsigned integers, with an overflow flag.
	 *
	 * _Available since v3.4._
	 */
	function tryAdd(uint256 a, uint256 b)
		internal
		pure
		returns (bool, uint256)
	{
		uint256 c = a + b;
		if (c < a) return (false, 0);
		return (true, c);
	}

	/**
	 * @dev Returns the substraction of two unsigned integers, with an overflow flag.
	 *
	 * _Available since v3.4._
	 */
	function trySub(uint256 a, uint256 b)
		internal
		pure
		returns (bool, uint256)
	{
		if (b > a) return (false, 0);
		return (true, a - b);
	}

	/**
	 * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
	 *
	 * _Available since v3.4._
	 */
	function tryMul(uint256 a, uint256 b)
		internal
		pure
		returns (bool, uint256)
	{
		// Gas optimization: this is cheaper than requiring 'a' not being zero, but the
		// benefit is lost if 'b' is also tested.
		// See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
		if (a == 0) return (true, 0);
		uint256 c = a * b;
		if (c / a != b) return (false, 0);
		return (true, c);
	}

	/**
	 * @dev Returns the division of two unsigned integers, with a division by zero flag.
	 *
	 * _Available since v3.4._
	 */
	function tryDiv(uint256 a, uint256 b)
		internal
		pure
		returns (bool, uint256)
	{
		if (b == 0) return (false, 0);
		return (true, a / b);
	}

	/**
	 * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
	 *
	 * _Available since v3.4._
	 */
	function tryMod(uint256 a, uint256 b)
		internal
		pure
		returns (bool, uint256)
	{
		if (b == 0) return (false, 0);
		return (true, a % b);
	}

	/**
	 * @dev Returns the addition of two unsigned integers, reverting on
	 * overflow.
	 *
	 * Counterpart to Solidity's `+` operator.
	 *
	 * Requirements:
	 *
	 * - Addition cannot overflow.
	 */
	function add(uint256 a, uint256 b) internal pure returns (uint256) {
		uint256 c = a + b;
		require(c >= a, "SafeMath: addition overflow");
		return c;
	}

	/**
	 * @dev Returns the subtraction of two unsigned integers, reverting on
	 * overflow (when the result is negative).
	 *
	 * Counterpart to Solidity's `-` operator.
	 *
	 * Requirements:
	 *
	 * - Subtraction cannot overflow.
	 */
	function sub(uint256 a, uint256 b) internal pure returns (uint256) {
		require(b <= a, "SafeMath: subtraction overflow");
		return a - b;
	}

	/**
	 * @dev Returns the multiplication of two unsigned integers, reverting on
	 * overflow.
	 *
	 * Counterpart to Solidity's `*` operator.
	 *
	 * Requirements:
	 *
	 * - Multiplication cannot overflow.
	 */
	function mul(uint256 a, uint256 b) internal pure returns (uint256) {
		if (a == 0) return 0;
		uint256 c = a * b;
		require(c / a == b, "SafeMath: multiplication overflow");
		return c;
	}

	/**
	 * @dev Returns the integer division of two unsigned integers, reverting on
	 * division by zero. The result is rounded towards zero.
	 *
	 * Counterpart to Solidity's `/` operator. Note: this function uses a
	 * `revert` opcode (which leaves remaining gas untouched) while Solidity
	 * uses an invalid opcode to revert (consuming all remaining gas).
	 *
	 * Requirements:
	 *
	 * - The divisor cannot be zero.
	 */
	function div(uint256 a, uint256 b) internal pure returns (uint256) {
		require(b > 0, "SafeMath: division by zero");
		return a / b;
	}

	/**
	 * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
	 * reverting when dividing by zero.
	 *
	 * Counterpart to Solidity's `%` operator. This function uses a `revert`
	 * opcode (which leaves remaining gas untouched) while Solidity uses an
	 * invalid opcode to revert (consuming all remaining gas).
	 *
	 * Requirements:
	 *
	 * - The divisor cannot be zero.
	 */
	function mod(uint256 a, uint256 b) internal pure returns (uint256) {
		require(b > 0, "SafeMath: modulo by zero");
		return a % b;
	}

	/**
	 * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
	 * overflow (when the result is negative).
	 *
	 * CAUTION: This function is deprecated because it requires allocating memory for the error
	 * message unnecessarily. For custom revert reasons use {trySub}.
	 *
	 * Counterpart to Solidity's `-` operator.
	 *
	 * Requirements:
	 *
	 * - Subtraction cannot overflow.
	 */
	function sub(
		uint256 a,
		uint256 b,
		string memory errorMessage
	) internal pure returns (uint256) {
		require(b <= a, errorMessage);
		return a - b;
	}

	/**
	 * @dev Returns the integer division of two unsigned integers, reverting with custom message on
	 * division by zero. The result is rounded towards zero.
	 *
	 * CAUTION: This function is deprecated because it requires allocating memory for the error
	 * message unnecessarily. For custom revert reasons use {tryDiv}.
	 *
	 * Counterpart to Solidity's `/` operator. Note: this function uses a
	 * `revert` opcode (which leaves remaining gas untouched) while Solidity
	 * uses an invalid opcode to revert (consuming all remaining gas).
	 *
	 * Requirements:
	 *
	 * - The divisor cannot be zero.
	 */
	function div(
		uint256 a,
		uint256 b,
		string memory errorMessage
	) internal pure returns (uint256) {
		require(b > 0, errorMessage);
		return a / b;
	}

	/**
	 * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
	 * reverting with custom message when dividing by zero.
	 *
	 * CAUTION: This function is deprecated because it requires allocating memory for the error
	 * message unnecessarily. For custom revert reasons use {tryMod}.
	 *
	 * Counterpart to Solidity's `%` operator. This function uses a `revert`
	 * opcode (which leaves remaining gas untouched) while Solidity uses an
	 * invalid opcode to revert (consuming all remaining gas).
	 *
	 * Requirements:
	 *
	 * - The divisor cannot be zero.
	 */
	function mod(
		uint256 a,
		uint256 b,
		string memory errorMessage
	) internal pure returns (uint256) {
		require(b > 0, errorMessage);
		return a % b;
	}
}

// File: contracts/openzeppelin/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.2 <0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: contracts/openzeppelin/token/ERC20/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "./IERC20.sol";
import "../../math/SafeMath.sol";
import "../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using SafeMath for uint256;
    using Address for address;

    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        // solhint-disable-next-line max-line-length
        require((value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).add(value);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 newAllowance = token.allowance(address(this), spender).sub(value, "SafeERC20: decreased allowance below zero");
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) { // Return data is optional
            // solhint-disable-next-line max-line-length
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: contracts/RollRewardsDistributionRecipient.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity =0.7.6;

import "./RollPausable.sol";

// https://docs.synthetix.io/contracts/RewardsDistributionRecipient
abstract contract RollRewardsDistributionRecipient is RollPausable {
	address public rewardsDistribution;

	function notifyRewardAmount(
		uint256[] calldata _rewards,
		address[] calldata _tokens
	) external virtual;

	modifier onlyRewardsDistribution() {
		_onlyRewardsDistribution();
		_;
	}

	function _onlyRewardsDistribution() private view {
		require(
			msg.sender == rewardsDistribution,
			"Caller is not RewardsDistribution contract"
		);
	}

	function setRewardsDistribution(address _rewardsDistribution)
		external
		onlyOwner
	{
		require(_rewardsDistribution != address(0), "Invalid address");
		rewardsDistribution = _rewardsDistribution;
	}
}

// File: contracts/TokenWrapper.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity =0.7.6;

import "./openzeppelin/token/ERC20/IERC20.sol";
import "./openzeppelin/math/SafeMath.sol";
import "./openzeppelin/token/ERC20/SafeERC20.sol";

contract TokenWrapper {
	using SafeMath for uint256;
	using SafeERC20 for IERC20;

	IERC20 public token;

	uint256 internal _totalSupply;
	mapping(address => uint256) internal _balances;

	function totalSupply() public view returns (uint256) {
		return _totalSupply;
	}

	function balanceOf(address account) public view returns (uint256) {
		return _balances[account];
	}

	function stake(uint256 amount) public virtual {
		_totalSupply = _totalSupply.add(amount);
		_balances[msg.sender] = _balances[msg.sender].add(amount);
		token.safeTransferFrom(msg.sender, address(this), amount);
	}

	function withdraw(uint256 amount) public virtual {
		_totalSupply = _totalSupply.sub(amount);
		_balances[msg.sender] = _balances[msg.sender].sub(amount);
		token.safeTransfer(msg.sender, amount);
	}
}

// File: contracts/RollPausable.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity =0.7.6;

import "./RollOwned.sol";

// https://docs.synthetix.io/contracts/Pausable
abstract contract RollPausable is RollOwned {
	uint256 public lastPauseTime;
	bool public paused;

	constructor() {
		// This contract is abstract, and thus cannot be instantiated directly
		require(owner != address(0), "Owner must be set");
		// Paused will be false, and lastPauseTime will be 0 upon initialisation
	}

	/**
	 * @notice Change the paused state of the contract
	 * @dev Only the contract owner may call this.
	 */
	function setPaused(bool _paused) external onlyOwner {
		// Ensure we're actually changing the state before we do anything
		if (_paused == paused) {
			return;
		}

		// Set our paused state.
		paused = _paused;

		// If applicable, set the last pause time.
		if (_paused) {
			lastPauseTime = block.timestamp;
		}

		// Let everyone know that our pause state has changed.
		emit PauseChanged(_paused);
	}

	event PauseChanged(bool isPaused);

	modifier notPaused {
		_notPaused();
		_;
	}

	function _notPaused() private view {
		require(
			!paused,
			"This action cannot be performed while the contract is paused"
		);
	}
}

// File: contracts/RollOwned.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity =0.7.6;

import "./StakingRegistry.sol";

// https://docs.synthetix.io/contracts/Owned
abstract contract RollOwned {
	address public owner;
	address public nominatedOwner;
	IStakingRegistry public immutable registry;

	constructor(address _owner, address _registry) {
		require(_owner != address(0), "Owner address cannot be 0");
		owner = _owner;
		registry = IStakingRegistry(_registry);
		emit OwnerChanged(address(0), _owner);
	}

	function nominateNewOwner(address _owner) external onlyOwner {
		nominatedOwner = _owner;
		emit OwnerNominated(_owner);
	}

	function acceptOwnership() external {
		require(
			msg.sender == nominatedOwner,
			"You must be nominated before you can accept ownership"
		);
		emit OwnerChanged(owner, nominatedOwner);
		registry.assignOwnerToContract(address(this), nominatedOwner, owner);
		owner = nominatedOwner;
		nominatedOwner = address(0);
	}

	modifier onlyOwner {
		_onlyOwner();
		_;
	}

	function _onlyOwner() private view {
		require(
			msg.sender == owner,
			"Only the contract owner may perform this action"
		);
	}

	event OwnerNominated(address newOwner);
	event OwnerChanged(address oldOwner, address newOwner);
}

// File: contracts/StakingRegistry.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity =0.7.6;

interface IStakingRegistry {
	function setCaller(address _caller, bool _value) external;

	function assignOwnerToContract(
		address _stakingContract,
		address _owner,
		address _previousOwner
	) external;
}

contract StakingRegistry is IStakingRegistry {
	address public controller;

	mapping(address => bool) public authorisedCallers;
	mapping(address => address) public contractToOwner;
	mapping(address => address[]) public ownerToContracts;
	mapping(address => uint256) public contractIndex;

	constructor() {
		controller = msg.sender;
	}

	event NewStakingContractOwner(
		address indexed stakingContract,
		address indexed newOwner,
		address indexed previousOwner
	);

	modifier onlyController() {
		_onlyController();
		_;
	}

	function _onlyController() private view {
		require(msg.sender == controller, "StakingRegistry: Not controller.");
	}

	modifier onlyCaller() {
		_onlyCaller();
		_;
	}

	function _onlyCaller() private view {
		require(
			authorisedCallers[msg.sender] || msg.sender == controller,
			"StakingRegistry: Unauthorised caller."
		);
	}

	function contractCountPerOwner(address _owner)
		external
		view
		returns (uint256)
	{
		return ownerToContracts[_owner].length;
	}

	function setController(address _newController) public onlyController {
		require(_newController != address(0), "Invalid address");
		controller = _newController;
	}

	function setCaller(address _caller, bool _value)
		public
		override
		onlyCaller
	{
		authorisedCallers[_caller] = _value;
	}

	function assignOwnerToContract(
		address _stakingContract,
		address _owner,
		address _previousOwner
	) public override onlyCaller {
		require(
			contractToOwner[_stakingContract] == _previousOwner,
			"Registry: previous owner is not owner of staking contract."
		);
		if (_previousOwner != address(0)) {
			address[] storage prevOwnerContracts =
				ownerToContracts[_previousOwner];
			(uint256 oldIndex, bool exists) =
				getIndexArray(prevOwnerContracts, _stakingContract);
			if (exists) {
				prevOwnerContracts[oldIndex] = prevOwnerContracts[
					prevOwnerContracts.length - 1
				];
				prevOwnerContracts.pop();
			}
		}
		uint256 index = ownerToContracts[_owner].length;
		contractToOwner[_stakingContract] = _owner;
		ownerToContracts[_owner].push(_stakingContract);
		contractIndex[_stakingContract] = index;
		emit NewStakingContractOwner(_stakingContract, _owner, _previousOwner);
	}

	function getIndexArray(address[] memory _array, address value)
		public
		pure
		returns (uint256, bool)
	{
		for (uint256 i = 0; i < _array.length; i++)
			if (value == _array[i]) return (i, true);
		return (0, false);
	}
}